clear all
set more off 
version 12.0
set trace off
cap log close
pause off

* Set path
if "`c(username)'"=="YOUR USER NAME"{
	global home "YOUR PATH TO REPLICATION FOLDER"
}
	global code "$home/Code"
	global data "$home/Data"
	global output "$home/Output"

set scheme s1color

* Load custom regression table formatting do files
include "$code/Balance_reg_fe.do"
include "$code/Balance_reg.do"

* Load data
use "$data/indivs_merge.dta", clear

/////////
// Figure S3.1 Mean adult work hours by treatment group over time
/////////
{
preserve
keep if inlist(wave,1,2,3,4)
keep if age>17
su workinghrs_l7 if trt_ctl==1 & t==3
local mean3= r(mean)
local se3=r(sd)/sqrt(r(N))
su workinghrs_l7 if trt_ctl==1 & t==9
local mean9= r(mean)
local se9=r(sd)/sqrt(r(N))

reghdfe workinghrs_l7 i.trt_ctl##ib3.t if wave<4, absorb(id i.month) vce(cluster id)
local beta1 = _b[1.trt_ctl#1.t]
local se1 = _se[1.trt_ctl#1.t]
local beta5 = _b[1.trt_ctl#5.t]
local se5 = _se[1.trt_ctl#5.t]
local beta7 = _b[1.trt_ctl#7.t]
local se7 = _se[1.trt_ctl#7.t]

keep if trt_ctl==0
collapse (mean) workl7=workinghrs_l7 , by(t)
gen workl7_trt=workl7+`beta1' if t==1
replace workl7_trt=`mean3' if t==3
replace workl7_trt=workl7+`beta5' if t==5
replace workl7_trt=workl7+`beta7' if t==7
replace workl7_trt=`mean9' if t==9
gen hi=workl7_trt + 1.962*`se1' if t==1
replace hi=workl7_trt + 1.962*`se3' if t==3
replace hi=workl7_trt + 1.962*`se5' if t==5
replace hi=workl7_trt + 1.962*`se7' if t==7
replace hi=workl7_trt + 1.962*`se9' if t==9
gen lo=workl7_trt - 1.962*`se1' if t==1
replace lo=workl7_trt - 1.962*`se3' if t==3
replace lo=workl7_trt - 1.962*`se5' if t==5
replace lo=workl7_trt - 1.962*`se7' if t==7
replace lo=workl7_trt - 1.962*`se9' if t==9
	
twoway 	(line workl7_trt t, color(black)) ///
		(rcap hi lo t , color(black)) ///
		(line workl7 t , color(gs8) lpattern(dash)) 		///
		,  ///	
		xtitle("", size(small)) ytitle("Total work hours in last 7 days", size(medsmall)) ///
		ylabel(0(5)30, labsize(medsmall) angle(0) nogrid) ///
		xlabel(1(2)10, valuelabels labsize(medsmall) angle(0)) ///
		xline(6, lcolor(red)) xline(8, lcolor(red))  ///
		text(29 3 "Schools fully" "closed" , color(black) size(medsmall) placement(s)) ///
		text(29 7 "Schools partially" "reopen" , color(black) size(medsmall) placement(s)) ///
		text(29 9 "Schools fully" "reopen" , color(black) size(medsmall) placement(s)) ///
		text(18 2 "Treatment" , color(black) size(medsmall) placement(n)) ///
		text(16 2 "Control" , color(gs8) size(medsmall) placement(s)) ///
		legend(off) 
restore
graph export "$output/Graphs/hours_mean_reg_adults.pdf",replace	
graph export "$output/Graphs/hours_mean_reg_adults.tif",replace	
}

/////////
// Table S3.1 Impact of partial reopening on adult labor supply in the last 7 days, by time period
/////////
{
preserve
keep if inlist(wave,1,2,3) 
keep if age>17

eststo clear
eststo wk: reghdfe working_l7 i.trt_ctl##ib3.t, absorb(id i.month) vce(cluster id)
test 1.trt_ctl#1.t 1.trt_ctl#3.t 1.trt_ctl#5.t
local Fp=round(`r(p)',0.001)
local fp1: display %4.3f `Fp'
su working_l7 if post==0 & trt_ctl==0 & e(sample)
estadd scalar Mean = r(mean)
estadd scalar fp=`fp1'
eststo wkh: reghdfe workinghrs_l7 i.trt_ctl##ib3.t, absorb(id i.month) vce(cluster id)
test 1.trt_ctl#1.t 1.trt_ctl#3.t 1.trt_ctl#5.t
local Fp=round(`r(p)',0.001)
local fp1: display %4.3f `Fp'
su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
estadd scalar Mean = r(mean)
estadd scalar fp=`fp1'

restore
esttab wk wkh using "$output/Tables/work_trend_coeffs.tex", ///
	keep(*.trt_ctl#*.t) ///
	scalars("Mean Mean, pre-reopen control" "fp p-value, pre-reopening coefs jointly 0 for treament") /// 
	mlabels("\shortstack{Engaged in any work \\ in last 7 days}" "\shortstack{Total work hours \\ in last 7 days}") ///		
	star(* 0.10 ** 0.05 *** 0.01) ///
	addnote("SEs clustered at household level.") ///
	se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs b(2) f end( \\ )
}
		
/////////
// Table S3.2 Impacts of partial school reopening on total working hours, varying controls
/////////	
{
preserve
keep if inlist(wave,1,2,3)
keep if age>17

global hhctl current_num_adults s2_q15c_youngchildren s2_q15b_schoolchildren ag_hh ent_hh resp_fem
global indctl age gender ishead

eststo clear
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local hfe = "Y"
	estadd local mfe = "Y"	
	estadd local cmfe2 = "N"	
	estadd local ife = "N"
	estadd local hhc = "N"
	estadd local indc = "N"
	quietly eststo: reghdfe workinghrs_l7 i.post##i.trt_ctl, absorb(id) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local hfe = "Y"
	estadd local mfe = "N"	
	estadd local cmfe2 = "N"	
	estadd local ife = "N"
	estadd local hhc = "N"
	estadd local indc = "N"	
	quietly eststo: reghdfe workinghrs_l7 i.posttreat $indctl, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local hfe = "Y"
	estadd local mfe = "Y"	
	estadd local cmfe2 = "N"	
	estadd local ife = "N"
	estadd local hhc = "N"
	estadd local indc = "Y"
	quietly eststo: reghdfe workinghrs_l7 i.posttreat $hhctl $indctl, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local hfe = "Y"
	estadd local mfe = "Y"	
	estadd local cmfe2 = "N"	
	estadd local ife = "N"
	estadd local hhc = "Y"
	estadd local indc = "Y"
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(uniqueindiv i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local hfe = "N"
	estadd local mfe = "Y"	
	estadd local cmfe2 = "N"	
	estadd local ife = "Y"
	estadd local hhc = "N"
	estadd local indc = "N"
	quietly eststo: reg workinghrs_l7 i.post##i.trt_ctl if e(sample), vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local hfe = "N"
	estadd local mfe = "N"	
	estadd local cmfe2 = "N"	
	estadd local ife = "N"
	estadd local hhc = "N"
	estadd local indc = "N"
	quietly eststo: reg workinghrs_l7 i.post##i.trt_ctl $hhctl $indctl if e(sample), vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local hfe = "N"
	estadd local mfe = "N"	
	estadd local cmfe2 = "N"	
	estadd local ife = "N"
	estadd local hhc = "Y"
	estadd local indc = "Y"
esttab using "$output/Tables/rob_main_ctls.tex", ///
		keep(1.post 1.post#1.trt_ctl 1.post#2.trt_ctl 1.posttreat 2.posttreat) ///		
		scalars("Mean Mean, pre-reopen control" "hfe Household FE" "ife Individual FE" "mfe Month FE" "indc Individual controls" "hhc Household controls") /// 
		star(* 0.10 ** 0.05 *** 0.01) ///
		mlabels(none) ///
		addnote("SEs clustered at household level.") ///
		b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )	
restore
}	

/////////
// Figure S3.2 Impacts of partial school reopening on total working hours, varying sample
/////////	
{
preserve
keep if inlist(wave,1,2,3)
keep if age>17
gen prer3=(wave==3 & post==0)
egen prer3max=max(prer3),by(id)

eststo clear
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Main}"
	quietly eststo: reghdfe workinghrs_l7 i.posttreat if prer3max==0, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Drop pre- \\ Oct 12 R3}"
	replace post=wave>2 // After mid-September
	replace posttreat=post*trt_ctl
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Post = \\ round 3}"
	replace post=today>22199 // After Oct 12
	replace posttreat=post*trt_ctl
	quietly eststo: reghdfe workinghrs_l7 i.posttreat if resp==1, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Respondent}"
	quietly eststo: reghdfe workinghrs_l7 i.posttreat if age<=65, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Age 18-65}"
	quietly eststo: reghdfe workinghrs_l7 i.posttreat if age>24 & age<=50, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Age 25-50}"
	quietly eststo: reghdfe workinghrs_l7 i.posttreat if parent_carer==1, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Potential \\ primary \\ caregivers}"
	quietly eststo: reghdfe workinghrs_l7 i.posttreat if trt_ctl!=2, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{No mixed \\ HHs}"
	replace posttreat=post*trt_ctl_no6
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Grade 6 \\ out of ctl}"
	replace posttreat=post*trt_ctl_add2
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Grade 2 \\ in ctl}"
	replace posttreat=post*trt_ctl_add2no9
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Grade 2 in, \\ Grade 9 \\ out of ctl}"
	replace posttreat=post*trt_ctl_full
	quietly eststo: reghdfe workinghrs_l7 i.posttreat, absorb(id i.month) vce(cluster id)
	su workinghrs_l7 if post==0 & trt_ctl==0 & e(sample)
	estadd scalar Mean = r(mean)
	estadd local sample = "\shortstack{Grade 12 \\ in trt, \\ Grades 10-11 \\ in ctl}"
restore
esttab using "$output/Tables/rob_main_sample.tex", ///
	keep(1.posttreat 2.posttreat) ///
	scalars("Mean Mean, pre-reopen control" "sample Sample") /// 
	mlabels(none) ///		
	star(* 0.10 ** 0.05 *** 0.01) ///
	addnote("SEs clustered at household level.") ///
	b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )

coefplot (est1, aseq(Main) ///
	\ est8, aseq(No mixed households) \ est2, aseq(Drop pre-Oct 12 R3 households) \ est3, aseq (Post defined by announcement) ///
	\ est4, aseq(Respondents only) \ est5, aseq (Adults age 18-65) ///
	\ est6, aseq(Adults age 25-50) \ est7, aseq (Potential primary caregivers) ///
	\ est9, aseq (Grade 6 out of control) ///
	\ est10, aseq(Grade 2 in control) \ est11, aseq (Grade 2 in, grade 9 out of control) ///
	\ est12, aseq(Grade 12 in treat, grades 10-11 in control)), ///
	keep(1.posttreat) xline(0) levels(95) nokey msymbol(T) mcolor(gs8) ciopts(lcolor(gs8)) byopts(xrescale) ///
	swapnames 
graph export "$output/Graphs/rob_main_sample.pdf",replace	
graph export "$output/Graphs/rob_main_sample.tif",replace	
}

/////////
// Table S3.3 Correlates of adult childcare hours in last 7 days
/////////
{
preserve
keep if wave==4 | wave==3
eststo clear
eststo: reg childcare_hr i.(gender resp ishead)##i.kidcount if age>17 & kidcount>0, cluster(id)
	su childcare_hr if gender==0 & resp==0 & ishead==0 & kidcount==1
	estadd scalar Mean = r(mean)
esttab using "$output/Tables/childcarehrs_demog.tex", ///
	drop(_cons) ///
	scalars("Mean Mean, Male non-resp non-head 0 kids") /// 
	star(* 0.10 ** 0.05 *** 0.01) ///
	addnote("SEs clustered at household level.") ///
	se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs b(2) f end( \\ )
}

/////////
// Table S3.4 Heterogeneity in effects on adult total work hours in the last 7 days, by sex
/////////
{
preserve
keep if inlist(wave,1,2,3) 
keep if age>17

eststo clear
foreach k of varlist workinghrs_l7 nonagri_hrs  {
	foreach z of varlist only_child young_kids trt_girl anygrade8 anychildag wealth_0 {
		quietly reghdfe `k' i.posttreat##i.`z' i.month##i.`z' if gender==0, absorb(id i.month) vce(cluster id)
		eststo: xlincom (Z0T=1.posttreat) (Z0M=2.posttreat) (Z1T=1.posttreat + 1.posttreat#1.`z') (Z1M=2.posttreat+ 2.posttreat#1.`z'), post
		test Z0T=Z1T
		local p1: display %4.3f `r(p)'
		estadd scalar p1= `p1'
		test Z0M=Z1M
		local p2: display %4.3f `r(p)'
		estadd scalar p2= `p2'
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
		quietly reghdfe `k' i.posttreat##i.`z' i.month##i.`z' if gender==1, absorb(id i.month) vce(cluster id)
		eststo: xlincom (Z0T=1.posttreat) (Z0M=2.posttreat) (Z1T=1.posttreat + 1.posttreat#1.`z') (Z1M=2.posttreat+ 2.posttreat#1.`z'), post
		test Z0T=Z1T
		local p1: display %4.3f `r(p)'
		estadd scalar p1= `p1'
		test Z0M=Z1M
		local p2: display %4.3f `r(p)'
		estadd scalar p2= `p2'
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
	}
}
restore
esttab est1 est2 est3 est4 est5 est6 est7 est8 est9 est10 est11 est12 using "$output/Tables/het_workhrs_sex.tex", ///
	scalars("Mean Mean, pre-reopen control" "p1 p-value Treated Z0=Z1" "p2 p-value Mixed Z0=Z1") /// 
	mlabels("\shortstack{Men}" "\shortstack{Women}" "\shortstack{Men}" "\shortstack{Women}" "\shortstack{Men}" "\shortstack{Women}" "\shortstack{Men}" "\shortstack{Women}" "\shortstack{Men}" "\shortstack{Women}" "\shortstack{Men}" "\shortstack{Women}" ) ///		
	star(* 0.10 ** 0.05 *** 0.01) ///
	addnote("SEs clustered at household level.") ///
	b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )
}

/////////
// S4 Tables for mechanisms regressions
/////////	
{	
preserve
keep if inlist(wave,1,2,3) 
keep if age>17

su anychildag wealth_0 private_skl young_kids only_child trt_girl anygrade8 if wave==3 & resp==1 & !missing(trt_ctl)
su trt_girl if wave==3 & resp==1 & trt_ctl==1
ta grade4_hh trt_ctl if wave==3 & resp==1
ta grade8_hh trt_ctl if wave==3 & resp==1
replace childcare_hr=0 if missing(childcare_hr)
replace hired_ag=0 if ag_hh==0

eststo clear
foreach k of varlist workinghrs_l7 nonagri_hrs {
quietly eststo: reghdfe `k' i.posttreat, absorb(id i.month) vce(cluster id)
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
}
foreach k of varlist childcare_hr s4_q2_hours_allchild {
		quietly eststo: reghdfe `k' i.posttreat if resp==1, absorb(id i.month) vce(cluster id)
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
}
foreach k of varlist workinghrs_l7 nonagri_hrs {
foreach z of varlist only_child young_kids trt_girl anygrade8 anychildag wealth_0 {
		quietly reghdfe `k' i.posttreat##i.`z' i.month##i.`z', absorb(id i.month) vce(cluster id)
		eststo: xlincom (Z0T=1.posttreat) (Z0M=2.posttreat) (Z1T=1.posttreat + 1.posttreat#1.`z') (Z1M=2.posttreat+ 2.posttreat#1.`z'), post
		test Z0T=Z1T
		local p1: display %4.3f `r(p)'
		estadd scalar p1= `p1'
		test Z0M=Z1M
		local p2: display %4.3f `r(p)'
		estadd scalar p2= `p2'
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
	}
}
foreach k of varlist childcare_hr s4_q2_hours_allchild {
foreach z of varlist only_child young_kids trt_girlv anygrade8 anychildagpre wealth_0 {
		quietly reghdfe `k' i.posttreat##i.`z' i.month##i.`z' if resp==1, absorb(id i.month) vce(cluster id)
		eststo: xlincom (Z0T=1.posttreat) (Z0M=2.posttreat) (Z1T=1.posttreat + 1.posttreat#1.`z') (Z1M=2.posttreat+ 2.posttreat#1.`z'), post
		test Z0T=Z1T
		local p1: display %4.3f `r(p)'
		estadd scalar p1= `p1'
		test Z0M=Z1M
		local p2: display %4.3f `r(p)'
		estadd scalar p2= `p2'
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
	}
}

local k hired_ag
quietly eststo: reghdfe `k' i.posttreat if resp==1, absorb(id i.month) vce(cluster id)
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
foreach z of varlist only_child young_kids trt_girlv anygrade8 anychildag wealth_00 {
quietly reghdfe `k' i.posttreat##i.`z' i.month##i.`z' if resp==1, absorb(id i.month) vce(cluster id)
eststo: xlincom (Z0T=1.posttreat) (Z0M=2.posttreat) (Z1T=1.posttreat + 1.posttreat#1.`z') (Z1M=2.posttreat+ 2.posttreat#1.`z'), post
		test Z0T=Z1T
		local p1: display %4.3f `r(p)'
		estadd scalar p1= `p1'
		test Z0M=Z1M
		local p2: display %4.3f `r(p)'
		estadd scalar p2= `p2'
		su `k' if post==0 & trt_ctl==0 & e(sample)
		estadd scalar Mean = r(mean)
}
restore

esttab est1 est5 est6 est7 est8 est9 est10 using "$output/Tables/het_wkhrs.tex", ///
	scalars("Mean Mean, pre-reopen control" "p1 p-value Treated Z0=Z1" "p2 p-value Mixed Z0=Z1") /// 
	mlabels("\shortstack{None}" "\shortstack{Only 1 \\ HH child}" "\shortstack{Any child \\ age 0-4}" "\shortstack{Any girl \\ in grades \\ 3-9}" "\shortstack{Any child \\ in grades \\ 7-9}" "\shortstack{Any child \\ in HH ag}"  "\shortstack{Above \\ mean HH \\ wealth}" ) ///		
	star(* 0.10 ** 0.05 *** 0.01) drop(_cons) ///
	addnote("SEs clustered at household level.") ///
	b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )
esttab est2 est11 est12 est13 est14 est15 est16 using "$output/Tables/het_nonaghrs.tex", ///
	scalars("Mean Mean, pre-reopen control" "p1 p-value Treated Z0=Z1" "p2 p-value Mixed Z0=Z1") /// 
	mlabels("\shortstack{None}" "\shortstack{Only 1 \\ HH child}" "\shortstack{Any child \\ age 0-4}" "\shortstack{Any girl \\ in grades \\ 3-9}" "\shortstack{Any child \\ in grades \\ 7-9}" "\shortstack{Any child \\ in HH ag}"  "\shortstack{Above \\ mean HH \\ wealth}" ) ///		
	star(* 0.10 ** 0.05 *** 0.01) drop(_cons) ///
	addnote("SEs clustered at household level.") ///
	b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )
esttab est3 est17 est18 est19 est20 est21 est22 using "$output/Tables/het_respcc.tex", ///
	scalars("Mean Mean, pre-reopen control" "p1 p-value Treated Z0=Z1" "p2 p-value Mixed Z0=Z1") /// 
	mlabels("\shortstack{None}" "\shortstack{Only 1 \\ HH child}" "\shortstack{Any child \\ age 0-4}" "\shortstack{Any girl \\ in grades \\ 3-9}" "\shortstack{Any child \\ in grades \\ 7-9}" "\shortstack{Any child \\ in HH ag}"  "\shortstack{Above \\ mean HH \\ wealth}" ) ///		
	star(* 0.10 ** 0.05 *** 0.01) drop(_cons) ///
	addnote("SEs clustered at household level.") ///
	b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )
esttab est4 est23 est24 est25 est26 est27 est28 using "$output/Tables/het_childaghrs.tex", ///
	scalars("Mean Mean, pre-reopen control" "p1 p-value Treated Z0=Z1" "p2 p-value Mixed Z0=Z1") /// 
	mlabels("\shortstack{None}" "\shortstack{Only 1 \\ HH child}" "\shortstack{Any child \\ age 0-4}" "\shortstack{Any girl \\ in grades \\ 3-9}" "\shortstack{Any child \\ in grades \\ 7-9}" "\shortstack{Any child \\ in HH ag}"  "\shortstack{Above \\ mean HH \\ wealth}" ) ///		
	star(* 0.10 ** 0.05 *** 0.01) drop(_cons) ///
	addnote("SEs clustered at household level.") ///
	b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )
esttab est29 est30 est31 est32 est33 est34 est35 using "$output/Tables/het_hiredag.tex", ///
	scalars("Mean Mean, pre-reopen control" "p1 p-value Treated Z0=Z1" "p2 p-value Mixed Z0=Z1") /// 
	mlabels("\shortstack{None}" "\shortstack{Only 1 \\ HH child}" "\shortstack{Any child \\ age 0-4}" "\shortstack{Any girl \\ in grades \\ 3-9}" "\shortstack{Any child \\ in grades \\ 7-9}" "\shortstack{Any child \\ in HH ag}"  "\shortstack{Above \\ mean HH \\ wealth}" ) ///		
	star(* 0.10 ** 0.05 *** 0.01) drop(_cons) ///
	addnote("SEs clustered at household level.") ///
	b(2) se(2) varwidth(45) num nonotes label replace booktabs nobaselevels obs f end( \\ )
}


